National Aeronautics and Space Administration
Goddard Earth Science
Data Information and Services Center (GES DISC)

README Document for the Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) Level 3 Gridded Retrieval Temperature Data and Level 3 Zonal Means Composition Data

SAMSN7L3GRIDT
SAMSN7L3ZMTG

Last Revised 10/02/2020

Goddard Earth Sciences Data and Information Services Center (GES DISC)
https://disc.gsfc.nasa.gov
NASA Goddard Space Flight Center
Code 610.2
Greenbelt, MD 20771 USA

Prepared By:

James E. Johnson	10/02/2020
Name GES DISC GSFC Code 610.2	Date

Reviewed By:

Name	mm/dd/yyyy
Name GSFC Code xxx	Date

Name	mm/dd/yyyy
Name GSFC Code xxx	Date

Goddard Space Flight Center
Greenbelt, Maryland

Revision History

Revision Date	***Changes***	***Author***
10/02/2020	Original	James E. Johnson

Table of Contents

1. Introduction

This document provides basic information on using the Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) Level-3 Gridded Retrieval Temperature Data (GRID-T) and the Level-3 Zonal Means Composition Data (ZMT-G) products.

1.1 Data Product Description

GRID-T:
The Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) Level-3 Gridded Retrieval Temperature Data product contains atmospheric temperature profiles at 62 retrieved pressure levels, zonal means and climatology values on a regular 2.5 x 10 degree lat-lon grid, as well as temperature and their errors at 10 pressure levels 100, 30, 10, 3, 1, 0.3, 0.1, 0.03, 0.01 and 0.003 mbar.

Each file contains one days worth of data. SAMS is a limb profiler and spatial coverage is near global between latitude -50 and +70 degrees. Vertical coverage is from 246 to 0.00124 mbar at 62 equal logarithmic spaced pressure levels. The data are available from 24 December 1978 through 9 June 1983. The principal investigators for the SAMS experiment were Prof. John T. Houghton and Dr. Fredric W. Taylor from Oxford University.

The GRID-T product was previously available from the NASA National Space Science Data Center (NSSDC) under the name SAMS Gridded Retrieval Temperature Data with the identifier ESAD-00016 (old id 78-098A-02B).

ZMT-G:
The Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) Level-3 Zonal Means Composition Data product contains mixing ratios of nitrous oxide (N_2O) and methane (CH_4) separated into daytime and nighttime measurements. Because the methane and the nitrous oxide channels cannot function simultaneously, only one type of measurement is made for any nominal day.

Each file contains three years of data at daily resolution. SAMS is a limb profiler and spatial coverage is near global between latitude -50 and +70 degrees. Vertical coverage is from 50 to 0.125 mbar at 31 equal logarithmic spaced pressure levels. The data are available from 1 January 1979 through 30 December 1981. The principal investigators for the SAMS experiment were Prof. John T. Houghton and Dr. Fredric W. Taylor from Oxford University.

This ZMT-G product was previously available from the NASA National Space Science Data Center (NSSDC) under the name SAMS Zonal Means Composition Data with the identifier ESAD-00180 (old id 78-098A-02C).

1.1.1 The Stratospheric and Mesospheric Sounder

The objective of the Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) experiment was to observe emission from the limb of the atmosphere through seven pressure-modulated cells and six detectors in order to determine temperature and vertical concentrations of H2O, N2O, CH4, CO, and NO in the stratosphere and mesosphere. Measurements of zonal wind in this region were attempted by observing the Doppler shift of atmospheric emission lines. Radiation from the limb of the atmosphere was incident on a scan mirror in front of a 15-cm aperture telescope. The scan mirror scanned the limb, viewed space for calibration, and viewed the atmosphere obliquely to obtain vertical profiles at three adjacent fields of view (100 km by 10 km at the limb). The instrument operated successfully from October 1978 until June 1983. The experiment was based on techniques used on the previously flown Selective Chopper Radiometers (SCR) on Nimbus 4 and 5, and the Pressure Modulated Radiometer (PMR) flown on Nimbus 6. The Improved Stratospheric and Mesospheric Sounder (ISAMS) was flown on the Upper Atmosphere Research Satellite in 1991.

1.1.2 Nimbus-7 Overview

The Nimbus-7 satellite was successfully launched on October 24, 1978 and was the final in the Nimbus series. The spacecraft included nine experiments: (1) the Limb Infrared Monitor of the Stratosphere (LIMS) for making vertical profiles of temperature and concentrations of O_3, H_2O, NO_2, and HNO_3, (2) a Stratospheric and Mesospheric Sounder (SAMS) providing vertical concentrations of H_2O, CH_4, CO and NO and measure the temperature in the upper atmosphere, (3) the Coastal-Zone Color Scanner (CZCS) for mapping ocean chlorophyll concentrations, (4) the Stratospheric Aerosol Measurement II (SAM II) to map the concentration and optical properties of aerosols, (5) the Earth Radiation Budget (ERB) for measuring the incoming and outgoing reflected and emitted radiation of the Earth, (6) a Scanning Multichannel Microwave Radiometer (SMMR) to obtain and use ocean momentum and energy-transfer parameters on a nearly all-weather operational basis., (7) a Solar Backscatter UV (SBUV) spectrometer to determine the vertical distribution of ozone, (8) the Total Ozone Mapping Spectrometer (TOMS) for mapping the total column amount of ozone, and (9) the Temperature Humidity Infrared Radiometer (THIR) for measuring daytime and nighttime surface and cloudtop temperatures, as well as the water vapor content of the upper atmosphere.

The orbit of the satellite can be characterized by the following:

- circular orbit at ~950 km
- inclination of 99 degrees
- period of an orbit is about 104 minutes
- orbits cross the equator at 26 degrees of longitude separation
- sun-synchronous

1.2 Algorithm Background

Both Nimbus-7 SAMS L3 data products were generated from the radiance archive tape data. The data were originally processed on IBM 360 computers using a 16-bit words, and copied to 1600 bpi 9-track tapes for archival. Further information on the SAMS instrument and data processing can be found in the Nimbus-7 Users' Guide Section 6 and in the Nimbus-7 SAMS Experiment User's Guide.

1.3 Data Disclaimer

The data should be used with care and one should first read the Nimbus-7 User's Guide, Section 6 describing the SAMS experiment, and the "Nimbus-7 SAMS Experiment User's Guide". Users should cite this data product in their research.

2. Data Organization

The Nimbus-7 Stratospheric and Mesospheric Sounder Level-3 Gridded Retrieval Temperature Data spans the time period from December 24, 1978 to June 9, 1983. Each file contains one day of data. The Nimbus-7 Stratospheric and Mesospheric Sounder Level-3 Zonal Means Composition Data spans the three year period from January 1, 1979 to December 30, 1981 in a single file at daily resolution.

2.1 File Naming Convention

The data product files are named according to the following convention:

<Platform>-<Instrument>_<Level>-<Product>_<DateTime>_<TapeNumber>.<Suffix>

where:
o Platform = name of the platform or satellite (Nimbus7)
o) Instrument = name of the instrument and product (SAMS)
o) Level = process level (L3)
o) Product = gridded retrieval temperature data (GRIDT or ZMTG)
o) Date = Data start date in format <YYYY>m<MMDD> where

1. YYYY = 4 digit year (1978 – 1979)
2. MM = 2 digit month (01-12)
3. DD = 2 digit day of month (01-31)

(for ZMT-G files the date is a range <begin date>-<end_date>, with dates using same format above)
o TapeNumber = 5 digit number of tape (preceded by 'DD' - primary or 'DC' - backup)
o Suffix = the file format (always TAP, indicating tape binary data)

```
File name examples: Nimbus7-SAMS_L3-GRIDT_1978m1224_DD65018.TAP
                    Nimbus7-SAMS_L3-ZMTG_1979m0101-1981m1230_DD65022.TAP
```

2.2 File Format and Structure

The data are stored as they were originally written in IBM binary (big-endian) record oriented structured files. The files were written on the original 1600 bpi 9-track tapes using a blocked FORTRAN format. The first file on the tape is the tape header file with two records containing BCD text encoded information about the tape. This is followed by a set of daily data files. Each daily file on the tape contains a set of records with a FORTRAN record size word, the record block, and a FORTRAN record trailing size word.

For GRID-T files there are 69 records in a daily file. The first record is the file header with a size of 20 bytes, the next 48 records are the gridded temperature data at 62 pressure levels with a size of 4268 bytes. The last 20 records contain alternating temperature and error values at 10 select pressure levels each with a size of 3268 bytes. At the end of the file there is an End-of-File word (the last file on the tape will end with a double End-of-File word). The data measurement records will contain the

temperature profiles. All data use big-endian 16-bit words. For the contents and layout of the documentation, see section 3.1.

During data recovery a total of 1541 daily data files were retrieved from the tapes, with 1113 files from 4 primary DD tapes, and just 428 files from 4 backup DC tapes. All of the files from the DC tapes were either identical copies of files from the DD tapes (362 files), contained missing records which were complete on the DD tape file counterparts (7 files), or contained duplicate records which were clean on the DD tape file counterparts (59 files). There were no files on the DC tapes that were not on a DD tape. In summary, the 1113 files from the DD tapes were clean and represent the complete record of the SAMS gridded retrieval temperature data collection and are archived at the GES DISC.

For the ZMT-G data, all days for the mission are contained in a single file. There are a total of 783 data records in the file, each record is of size 5986 bytes and represents a single day and may contain either nitrous oxide (record type 7405) or methane data (record type 7406). All data are stored using big-endian 16-bit words. For the contents and layout see section 3.1.

During data recovery a total of one file was retrieved from one primary DD tape, and one file from one backup DC tape. Both files were found to be exactly identical. Therefore only the one ZMT-G file from the primary DD tape is archived at the GES DISC.

2.3 Key Science Data Fields

The primary science data field in the GRID-T data product is the SAMS temperature from the 62 pressure levels. The primary science data field in the ZMT-G data product are the N2O and CH4 zonal means at 31 pressure levels.

Figure 1: Typical data coverage for a Nimbus-7 SAMS Level 3 GRID-T daily data file showing temperatures at 10 of the 62 pressure levels for Dec. 24, 1978.



Figure 2: Data from Nimbus-7 SAMS Level 3 ZMT-G showing mixing ratios of N_2O (Jan. 2, 1979) and CH_4 (Jan. 3, 1979). SAMS can only measure in one channel per day, hence the two different dates.



3. Data Contents

The granularity for the GRID-T product is one day, and for the ZMT-G product it is the entire mission with daily resolution.

3.1 Data Records

The Nimbus-7 User's Guide does not describe the layout of the file format. Refer instead to Chapter 6 in the NASA Reference Publication 1221, titled "Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) Experiment Data User's Guide".

The original tape contained about a years worth of data files for GRID-T, and in the case of ZMT-G the entire mission. A tape would have two tape headers, followed by the daily data files with zero byte word End-of-File (EOF) markers in between and ending with a double zero byte End-of-Tape (EOT) marker.

The GRID-T tapes each included a tape header file. These were then followed by a set of daily data files, which would typically include a header records, then 48 records containing temperatures at each retrieved pressure level, followed by 20 records containing the temperature and error at 10 standard pressure levels. At the end there is a 0 byte EOF word. As part of the recovery, the GES DISC has extracted and archived the daily data files from the tapes. The original data were written on IBM machines using big-endian 16-bit words. Each record type is shown in the tables 3-1 to 3-3 below.

The ZMT-G tapes include a tape header file. This was followed by a single file containing the entire 3 year record, where each 5986 byte record contains a single day of data. Two 0-byte EOT words and the tape. The original data was written on IBM machines using big-endian 16-bit words. Each record type is shown in the tables 3-4.

Table 3-1 Header, Record Type = 7400

16-bit Word	Field Name	Units	Size	Comments
1	Record Length (in bytes)	-	1	
2	Serial Number	-	1	
3	Record Type	-	1	7400
4	Data File Number	-	1	
5	Data Year	-	1	
6	Data Day	-	1	
7	7401	-	1	
8	7402	-	1	
9	7403	-	1	
10	Zero Fill	-	1	
11	Checksum	-	1	
12 – 20	Spares	-	9	Set to zero

Table 3-2: Temperature at All 62 Pressure Levels Data, Record Type = 7402

16-bit Word	Field Name	Units	Size	Comments
1	Record Length (in bytes)	-	1	
2	Serial Number	-	1	
3	Record Type	-	1	7402
4	Data Day	-	1	
5	Data Year	-	1	
6	Processing Day	-	1	
7	Processing Year	-	1	
8	Grid Point Latitude x 100	degrees	1	
9	Grid Point Longitude x 100	degrees	1	-180
10 – 71	Temperature x 100 for 62 Pressure Levels	deg K	62	(see note below)
72	Grid Point Latitude x 100	degrees	1	
73	Grid Point Longitude x 100	degrees	1	-170

74 – 135	Temperature x 100 for 62 Pressure Levels	deg K	62	
⋮	⋮	⋮	⋮	⋮
2248	Grid Point Latitude x 100	degrees	1	
2249	Grid Point Longitude x 100	degrees	1	+170
2250 – 2311	Temperature x 100 for 62 Pressure Levels	deg K	62	
2312	Grid Point Latitude x 100	degrees	1	
2313	Grid Point Longitude x 100	degrees	1	+190
2314 – 2375	Zonal Mean x 100 for 62 Pressure Levels	deg K	62	
2376	Grid Point Latitude x 100	degrees	1	
2377	Grid Point Longitude x 100	degrees	1	+200
2378 – 2439	Climatology x 100 for 62 Pressure Levels	deg K	62	
2440	Checksum	-	1	2 byte word
2441	Zero Fill	-	1	

Note: 62 pressure levels starting at $\ln(p_0/p) = 1.4$, every 0.2 in $\ln(p_0/p)$ up to 13.6 (p_0=1000 mbar)

Table 3-3: Temperature and Error at 10 Standard Pressure Levels Data, Record Type = 7403

Word	Field Name	Units	Size	Comments
1	Record Length (in bytes)	-	1	
2	Serial Number	-	1	
3	Record Type	-	1	7403
4	Measurement Type	-	1	3
5	Data Day	-	1	
6	Data Year	-	1	
7	Processing Day	-	1	
8	Processing Year	-	1	
9	Ignore	-	1	
10	Scale Factor	-	1	-180
11	Data Type	-	62	2 for temperature 102 for error

12	Pressure Level (stored as 1000*ln(pressure/1000))	mbar	1	(see note below)
13	Ignore	-	3	
14	Ignore	-	1	
15	Ignore	-	1	
16 – 23	Zero Fill	-	7	
24 – 1751	Temperature Data or Error Array A(I,J) is for (J-21)*2.5°N, (I-19)*10°E	deg K	36 x 48	-32768 fill value
1752	Checksum	-	1	
1753	Zero Fill	-	1	

Note: pressure in mbar (value in parentheses are the corresponding values of word12):
100 (2303),30 (3507),10 (4605),3 (5809),1 (6908), 3 (8812), 1 (9210), 0.3 (10414), 0.1 (11513), 0.003 (12717).

Table 3-4: Zonal Means Composition Data, Record Type = 7405 (N_2O) or 7406 (CH_4)

16-bit Word	Field Name	Units	Size	Comments
1	Record Length (in bytes)	-	1	
2	Serial Number	-	1	
3	Record Type	-	1	7405 or 7406
4	Data Day	-	1	
5	Data Year	-	1	
6	Enabled Channel	-	1	
7	Sieve Setting Enabled PMC	-	1	
8	Sieve Setting Clamped PMC	-	1	
9	Sieve Setting A1 PMC	-	1	
10	Sieve Setting C1 PMC	-	1	
11	Processing Day	-	1	
12	Processing Year	-	1	
13	Number of Elements in Profile	-	1	
14	Bottom Level	mbar	1	$10*\ln(P_0/P)$ see note
15	Top Level	mbar	1	$10*\ln(P_0/P)$ see note

16 – 46	Retrieved Mixing Ratio at -50.0° Latitude for 31 Pressure Levels	mixing ratio	31	see note below
47 – 77	Retrieved Mixing Ratio at -47.5° Latitude for 31 Pressure Levels	mixing ratio	31	see note below
⋮	⋮	⋮	⋮	⋮
1442 – 1472	Retrieved Mixing Ratio at +65.0° Latitude for 31 Pressure Levels	mixing ratio	31	(see note below)
1473 – 1503	Retrieved Mixing Ratio at +67.5° Latitude for 31 Pressure Levels	mixing ratio	31	(see note below)
1504 – 1534	Mixing Ratio Errors at -50.0° Latitude for 31 Pressure Levels	mixing ratio	31	(see note below)
1535 – 1565	Mixing Ratio Errors at -47.5° Latitude for 31 Pressure Levels	mixing ratio	31	(see note below)
⋮	⋮	⋮	⋮	⋮
2930 – 2960	Mixing Ratio Errors at +65.0° Latitude for 31 Pressure Levels	mixing ratio	31	(see note below)
2961 – 2991	Mixing Ratio Errors at +67.5° Latitude for 31 Pressure Levels	mixing ratio	31	(see note below)
2992	Checksum	-	1	
2993	Zero Fill	-	1	

Note: 31 pressure levels starting at $\ln(p_0/p) = 3.0$, every 0.2 in $\ln(p_0/p)$ up to 9.0 (p_0=1013.25 mbar)

3.2 Metadata

The metadata are contained in a separate XML formatted file having the same name as the data file with .xml appended to it.

Table 3-2: Metadata attributes associated with the data file.

Name	Description
LongName	Long name of the data product.
ShortName	Short name of the data product.
VersionID	Product or collection version.
GranuleID	Granule identifier, i.e. the name of the file.
Format	File format of the data file.
CheckSumType	Type of checksum used.
CheckSumValue	The value of the calculated checksum.
SizeBytesDataGranule	Size of the file or granule in bytes.
InsertDateTime	Date and time when the granule was inserted into the archive. The format for date is YYYY-MM-DD and time is hh-mm-ss.
ProductionDateTime	Date and time the file was produced in format YYYY-MM-DDThh:mm:ss.ssssssZ
RangeBeginningDate	Begin date when the data was collected in YYYY-MM-DD format.
RangeBeginningTime	Begin time of the date when the data was collected in hh-mm-ss format.
RangeEndingDate	End date when the data was collected in YYYY-MM-DD format.
RangeEndingTime	End time of the date when the data was collected in hh-mm-ss format.
PlatformShortName	Short name or acronym of the platform or satellite
InstrumentShortName	Short name or acronym of the instrument
SensorShortName	Short name or acronym of the sensor
WestBounding Coordinate	The westernmost longitude of the bounding rectangle(-180.0 to +180.0)
NorthBounding Coordinate	The northernmost latitude of the bounding rectangle(-90.0 to +90.0)
EastBounding Coordinate	The easternmost longitude of the bounding rectangle(-180.0 to +180.0)
SouthBounding Coordinate	The southernmost latitude of the bounding rectangle(-90.0 to +90.0)
ElapsedMinTime	Duration in minutes of data collected.

4. Reading the Data

The data are written in a binary record-oriented format. Using the record format specification in the section above, users can write software to read the data files. Please note that the data were originally written using a big-endian format, therefore users on little-endian machines will need to swap bytes for the words.

Two sample FORTRAN programs are included in the Appendix section which will read and print the the data contents. Additionally a FORTRAN function is included to perform byte swapping.

5. Data Services

5.1 GES DISC Search

The GES DISC provides a keyword, spatial, temporal and advanced (event) searches through its unified search and download interface:

https://disc.gsfc.nasa.gov/

5.2 Documentation

The data product landing pages provide information about these data products, as well as links to download the data files and relevant documentation:

https://disc.gsfc.nasa.gov/datacollection/SAMSN7L3GRIDT_001.html

https://disc.gsfc.nasa.gov/datacollection/SAMSN7L3ZMTG_001.html

5.3 Direct Download

These data products are available for users to download directly using HTTPS:

https://acdisc.gesdisc.eosdis.nasa.gov/data/Nimbus7_SAMS_Level3/SAMSN7L3GRIDT.001/

https://acdisc.gesdisc.eosdis.nasa.gov/data/Nimbus7_SAMS_Level3/SAMSN7L3ZMTG.001/

6. More Information

6.1 Contact Information

Name: GES DISC Help Desk
URL: https://disc.gsfc.nasa.gov/
E-mail: gsfc-help-disc@lists.nasa.gov
Phone: 301-614-5224
Fax: 301-614-5228
Address: Goddard Earth Sciences Data and Information Services Center
Attn: Help Desk
Code 610.2
NASA Goddard Space Flight Center
Greenbelt, MD 20771, USA

6.2 References

"The Nimbus-7 User's Guide - Section 6: The Stratospheric and Mesospheric Sounder (SAMS) Experiment", NASA Goddard Space Flight Center, November 1972, Pages 71-103

F. W. Taylor and S. T. Nutter, "Nimbus-7 Stratospheric and Mesospheric Sounder (SAMS) Experiment Data User's Guide", NASA Ref. Pub. 1221, NASA Goddard Space Flight Center, Greenbelt, MD, 1989

7. Appendices

Acknowledgments

The Nimbus data recovery task at the GES DISC is funded by NASA's Earth Science Data and Information System program.

Acronyms

BCD: Binary Coded Decimal

EOS: Earth Observing System

ESDIS: Earth Science and Data Information System

GES DISC: Goddard Earth Sciences Data and Information Services Center

GRID-T: Gridded Retrieved Temperature Tapes

GSFC: Goddard Space Flight Center

L3: Level-3 Gridded Data

SAMS: Stratospheric and Mesospheric Sounder

NASA: National Aeronautics and Space Administration

QA: Quality Assessment

UT: Universal Time

ZMT-G: Zonal Means Composition Data

FORTRAN Code

```
C-----------------------------------------------------------------------
C ^NAME: READ_SAMSL3
C   This program will read both the Nimbus 7 SAMS Level 3 Gridded
C   Retrieval Temperature Data (GRID-T) and the Zonal Mean Composition
C   Data (ZMT-G) files.
C
C   The GRID-T files contain a header (type 7400, size 40 bytes), then
C   48 temperature at 62 retrieval pressure level records (type 7402,
C   size 4882 bytes), and 20 temperature/standard error at 10 standard
C   pressure level records (type 7403, size 3506 bytes). The data are
C   gridded at 2.5 deg latitude and 10 deg longitude.

C   The ZMT-G files contain N2O mixing ratio records (type 7405, size
C   5986 bytes) and CH4 mixing ratio records (type 7406, size 5986
C   bytes). Data are in 2.5 deg latitude zones at 31 pressure levels.

C   This program will print the contents of each data record.
C
C ^MAJOR VARIABLES:
C     FNAME  - name of input file
C     BUFF   - buffer for data record
C     TEMP   - buffer for holding temporary 4-byte word
C     WORD   - integer 4-byte word
C     IBLKSZ - size of record block in bytes
C     IOS    - I/O status number
C
C ^NOTES:
C     Compile: gfortran -o READ_SAMSGRIDT.EXE READ_SAMSGRIDT.FOR
C
C ^ORGANIZATION: NASA/GSFC, Code 610.2
C
C ^AUTHOR:  James Johnson
C
C ^ADDRESS: james.johnson@nasa.gov
C
C ^CREATED: Sept 15, 2020
C-----------------------------------------------------------------------

      CHARACTER          FNAME*256      ! Filename
      CHARACTER          BUFF(21484)    ! Buffer for data record block
      INTEGER*4          IBLKSZ         ! Size of records
      INTEGER*4          IWORD          ! 4-byte word
      INTEGER*2          RECTYP         ! Record type
      INTEGER*2          I2SWAP         ! Function swaps short ints
      CHARACTER          TEMP(4)        ! Buffer to hold 4-byte word
      EQUIVALENCE        (TEMP,IWORD)

C Get the name of the input data file to read
      WRITE (0, *), 'Enter the name of the input file:'
      READ (5,'(A)') FNAME
      PRINT '("FILE = ",A)', FNAME
```

```
C Open the specified input file
      OPEN (UNIT=1, FILE=FNAME, STATUS='OLD', ACCESS='DIRECT',
     &      FORM='UNFORMATTED', RECL=1, ERR=99, IOSTAT=IOS)

C Initialize N (record number) and IOFF (byte offset in file)
      N=0
      IOFF=0

C Loop through the file reading all records in file
   5  DO

C Read the first 4-byte word or record size header
   10   DO I=1,4
          READ (1, REC=IOFF+I, IOSTAT=IOS, ERR=90) TEMP(I)
        END DO
        IBLKSZ = IWORD
        IOFF=IOFF+(I-1)

C End-of-File (EOF) mark, exit loop
        IF (IBLKSZ .EQ. 0) EXIT

C Next read the block of data
   20   DO I=1,IBLKSZ
          READ (1, REC=IOFF+I, IOSTAT=IOS) BUFF(I)
          IF (IOS .NE. 0) THEN
            PRINT '("ERROR: BUFF ",I4,X,I4,", IOSTAT: ",I6)', N,I-1,IOS
            STOP
          END IF
        END DO
        IOFF=IOFF+(I-1)
        N=N+1

C Split data records from record block
        RECTYP = I2SWAP(BUFF(5:6))
        IF      (RECTYP .EQ. 7400) THEN
          CALL PR7400(IBLKSZ,BUFF)
        ELSE IF (RECTYP .EQ. 7402) THEN
          CALL PR7402(IBLKSZ,BUFF)
        ELSE IF (RECTYP .EQ. 7403) THEN
          CALL PR7403(IBLKSZ,BUFF)
        ELSE IF (RECTYP .EQ. 7405 .OR. RECTYP .EQ. 7406) THEN
          CALL PRZMTG(IBLKSZ,BUFF)
        ELSE
          PRINT '("Record Type: Unknown")'
        END IF

C Finally read the last 4-byte word (should match first record size)
   30   DO I=1,4
          READ (1, REC=IOFF+I, IOSTAT=IOS, ERR=90) TEMP(I)
        END DO
        IF (IBLKSZ .NE. IWORD) THEN
          PRINT '("WARNING: IBLKSZ ",I10," != ",I10)', IBLKSZ, IWORD
        END IF
        IOFF=IOFF+(I-1)

      END DO
```

```
C Close the input file
   90 CLOSE(1)
      STOP

   99 PRINT '("ERROR: OPEN FILE, IOSTAT: ",I6)', IOS

  100 STOP
      END


C--------------------------------------------------------------------------
C     This Subroutine will print the 7400 GRID-T Records
C--------------------------------------------------------------------------
      SUBROUTINE PR7400(IBLKSZ,BUFF)

      CHARACTER     BUFF(40)                          ! Buffer for record block
      INTEGER*2     IWORD(20)                         ! Array of short ints
      INTEGER*2     I2SWAP                            ! Function swaps short ints

      DO I = 1,20
        IWORD(I) = I2SWAP(BUFF(2*I-1:2*I))
      END DO

      PRINT '("RECLEN =",X,I8)', IWORD(1)
      PRINT '("SERNUM =",X,I8)', IWORD(2)
      PRINT '("RECTYP =",X,I8)', IWORD(3)
      PRINT '("FILNUM =",X,I8)', IWORD(4)
      PRINT '("YEAR   =",X,I8)', IWORD(5)
      PRINT '("DAY    =",X,I8)', IWORD(6)
      PRINT '("RECTY1 =",X,I8)', IWORD(7)
      PRINT '("RECTY2 =",X,I8)', IWORD(8)
      PRINT '("RECTY3 =",X,I8)', IWORD(9)
C     PRINT '("FILL   =",X,I8)', IWORD(10)
      PRINT '("CHKSUM =",X,I8)', IWORD(11)
      PRINT '("=================")'

      RETURN
      END

C--------------------------------------------------------------------------
C     This Subroutine will print the 7402 GRID-T Records
C--------------------------------------------------------------------------
      SUBROUTINE PR7402(IBLKSZ,BUFF)

      CHARACTER     BUFF(4882)                        ! Buffer for record block
      INTEGER*2     IWORD(2441)                       ! Array of short ints
      INTEGER*2     I2SWAP                            ! Function swaps short ints

      DO I = 1,2441
        IWORD(I) = I2SWAP(BUFF(2*I-1:2*I))
      END DO

      PRINT '("RECLEN =",X,I8)', IWORD(1)
      PRINT '("SERNUM =",X,I8)', IWORD(2)
      PRINT '("RECTYP =",X,I8)', IWORD(3)
      PRINT '("YEAR   =",X,I8)', IWORD(4)
      PRINT '("DAY    =",X,I8)', IWORD(5)
```

```
      PRINT '("PRYEAR =",X,I8)', IWORD(6)
      PRINT '("PRDAY  =",X,I8)', IWORD(7)
      PRINT '("----------------")'
      DO I = 1,36
        PRINT '("TLAT   =",X,F8.2)', IWORD(64*(I-1)+8)/100.
        PRINT '("TLON   =",X,F8.2)', IWORD(64*(I-1)+9)/100.
        PRINT '("TVAL   =",/,10(X,F8.2))',IWORD(64*(I-1)+10:64*I+7)/100.
        PRINT '("-----------------")'
      END DO
      PRINT '("ZMLAT  =",X,F8.2)', IWORD(64*(I-1)+8)/100.
      PRINT '("ZMLON  =",X,F8.2)', IWORD(64*(I-1)+9)/100.
      PRINT '("ZMVAL  =",/,10(X,F8.2))',IWORD(64*(I-1)+10:64*I+7)/100.
      PRINT '("-----------------")'
      PRINT '("CLMLAT =",X,F8.2)', IWORD(64*I+8)/100.
      PRINT '("CLMLON =",X,F8.2)', IWORD(64*I+9)/100.
      PRINT '("CLMVAL =",/,10(X,F8.2))',IWORD(64*I+10:64*(I+1)+7)/100.
      PRINT '("-----------------")'
      PRINT '("CHKSUM =",X,I8)', IWORD(2440)
C     PRINT '("FILL   =",X,I8)', IWORD(2441)
      PRINT '("=================")'

      RETURN
      END


C-----------------------------------------------------------------------
C     This Subroutine will print the 7403 GRID-T Records
C-----------------------------------------------------------------------
      SUBROUTINE PR7403(IBLKSZ,BUFF)

      CHARACTER     BUFF(3506)                      ! Buffer for record block
      INTEGER*2     IWORD(1753)                     ! Array of short ints
      INTEGER*2     I2SWAP                          ! Function swaps short ints

      DO I = 1,1753
        IWORD(I) = I2SWAP(BUFF(2*I-1:2*I))
      END DO

      PRINT '("RECLEN =",X,I8)', IWORD(1)
      PRINT '("SERNUM =",X,I8)', IWORD(2)
      PRINT '("RECTYP =",X,I8)', IWORD(3)
      PRINT '("MEASMT =",X,I8)', IWORD(4)
      PRINT '("YEAR   =",X,I8)', IWORD(5)
      PRINT '("DAY    =",X,I8)', IWORD(6)
      PRINT '("PRYEAR =",X,I8)', IWORD(7)
      PRINT '("PRDAY  =",X,I8)', IWORD(8)
C     PRINT '("IGNORE =",X,I8)', IWORD(9)
      PRINT '("SCALE  =",X,I8)', IWORD(10)
      PRINT '("DTYPE  =",X,I8)', IWORD(11)
      PRINT '("LEVEL  =",X,I8)', IWORD(12)
C     PRINT '("IGNORE =",3(X,I8))', IWORD(13:15)
C     PRINT '("SPARE  =",8(X,I8))', IWORD(16:23)
      PRINT '("----------------")'
      IF      (IWORD(11) .EQ. 1) THEN
        PRINT'("COEF   =")'
      ELSE IF (IWORD(11) .EQ. 2) THEN
        PRINT'("TVAL   =")'
```

```
      ELSE IF (IWORD(11) .EQ. 101) THEN
        PRINT'("CERR   =")'
      ELSE IF (IWORD(11) .EQ. 102) THEN
        PRINT'("TERR   =")'
      END IF
      DO I = 1,36
        PRINT'(10(X,F8.2))',IWORD(48*(I-1)+24:48*I+23)/REAL(IWORD(10))
      END DO
      PRINT '("-----------------")'
      PRINT '("CHKSUM =",X,I8)', IWORD(1752)
C     PRINT '("FILL   =",X,I8)', IWORD(1753)
      PRINT '("=================")'

      RETURN
      END


C-----------------------------------------------------------------------
C     This Subroutine will print the 7405 and 7406 ZMT-G Records
C-----------------------------------------------------------------------
      SUBROUTINE PRZMTG(IBLKSZ,BUFF)

      CHARACTER     BUFF(5986)                      ! Buffer for record block
      INTEGER*2     IWORD(2993)                     ! Array of short ints
      INTEGER*2     I2SWAP                          ! Function swaps short ints

      DO I = 1,2993
        IWORD(I) = I2SWAP(BUFF(2*I-1:2*I))
      END DO

      PRINT '("RECLEN =",X,I7)', IWORD(1)
      PRINT '("SERNUM =",X,I7)', IWORD(2)
      PRINT '("RECTYP =",X,I7)', IWORD(3)
      PRINT '("DAY    =",X,I7)', IWORD(4)
      PRINT '("YEAR   =",X,I7)', IWORD(5)
      PRINT '("CHAN   =",X,I7)', IWORD(6)
      PRINT '("ENBPMC =",X,I7)', IWORD(7)
      PRINT '("CLMPMC =",X,I7)', IWORD(8)
      PRINT '("A1PMC  =",X,I7)', IWORD(9)
      PRINT '("C1PMC  =",X,I7)', IWORD(10)
      PRINT '("PRDAY  =",X,I7)', IWORD(11)
      PRINT '("PRYEAR =",X,I7)', IWORD(12)
      PRINT '("NELEM  =",X,I7)', IWORD(13)
      PRINT '("TOPLVL =",X,F7.1)', IWORD(14)/10.
      PRINT '("BTMLVL =",X,F7.1)', IWORD(15)/10.
      PRINT '("MIXING =")'
      DO I=1,48
        PRINT '("-- LAT ",F6.1," --")', (I-21)*2.5
        IF (IWORD(3) .EQ. 7405) THEN
          PRINT'(9(X,F7.2))',IWORD(16+(I-1)*31:15+(I)*31)/50.
        ELSE
          PRINT'(9(X,F7.4))',IWORD(16+(I-1)*31:15+(I)*31)/10000.
        END IF
      END DO
```

```
      PRINT '("ERROR  =")'
      DO I=1,48
        PRINT '("-- LAT ",F6.1," --")', (I-21)*2.5
        IF (IWORD(3) .EQ. 7405) THEN
          PRINT'(8(X,F8.2))',IWORD(1504+(I-1)*31:1503+(I)*31)/50.
        ELSE
          PRINT'(8(X,F8.4))',IWORD(1504+(I-1)*31:1503+(I)*31)/10000.
        END IF
      END DO
      PRINT '("----------------")'
      PRINT '("CHKSUM =",X,I7)', IWORD(2992)
C     PRINT '("FILL   =",X,I7)', IWORD(2993)
      PRINT '("================")'

      RETURN
      END


C-------------------------------------------------------------------------
C ^FUNCTION: I2SWAP
C
C     This function will swap the bytes of a 2-byte word
C-------------------------------------------------------------------------

      INTEGER*2 FUNCTION I2SWAP(BUFF)

      CHARACTER          BUFF(2)        ! Input data buffer
      CHARACTER          TEMP(2)        ! Output swapped buffer
      INTEGER*2          I2BUFF
      EQUIVALENCE        (TEMP, I2BUFF)

      TEMP(1) = BUFF(2)
      TEMP(2) = BUFF(1)
      I2SWAP = I2BUFF

      RETURN
      END
```